Mathematische Methoden in der Physik

Bonus Unterlagen : Aufgaben und Lösungen

Hinweis : Diese Texte wurden mit Hilfe von WOLFRAM MATHEMATICA 12 (Wolfram Research, Inc.) erstellt.

Viele in diesen Aufgaben gestellten Probleme können mit *Mathematica* direkt durch spezielle Befehle (wie etwa Limit oder Series) gelöst werden. Da es uns hier aber auf das entsprechende Hintergrundwissen ankommt, werden wir meist ausführlicher vorgehen und erst in späteren Rechnungen oder zur Kontrolle diese Befehle verwenden.

Kapitel 7. Grundlagen der Vektoranalysis

7.1

Zeigen Sie, dass

$d|\mathbf{v}|/dt \neq |\boldsymbol{b}|$.

Lösungsweg

Es ist $|\mathbf{v}| = \sqrt{\boldsymbol{v}.\boldsymbol{v}} = \sqrt{\sum_i \boldsymbol{v}_i^2}$ und daher

$$d|\boldsymbol{v}|/dt = d\sqrt{\sum_i \boldsymbol{v}_i^2}\,/dt = \sum_j (d\,\boldsymbol{v}_j/d\mathrm{t})\;(d/d\,\boldsymbol{v}_j)\sqrt{\sum_i \boldsymbol{v}_i^2}$$

$$= \sum_j (d\,\boldsymbol{v}_j/d\mathrm{t})\;\boldsymbol{v}_j\Big/\sqrt{\sum_i \boldsymbol{v}_i^2} = \frac{\dot{\boldsymbol{v}}\boldsymbol{.}\mathrm{v}}{|\mathbf{v}|}\;.$$

Das ist die Komponente von $\dot{\boldsymbol{v}}$ in die Richtung von v, wohingegen $|\boldsymbol{b}| = |\dot{\boldsymbol{v}}|$.

```
In[∘]:= ClearAll["Global`*"];
```

7.2

Der Vektor $\boldsymbol{u} = \boldsymbol{a}\cos t + \boldsymbol{b}\sin t$ wird mit Hilfe konstanter Vektoren $\boldsymbol{a}$ und $\boldsymbol{b}$ gebildet. Man zeige

$\boldsymbol{u}\;.\;[(d\boldsymbol{u}/dt)\times(d^2\boldsymbol{u}/d\,t^2)] = 0.$

Lösungsweg

Wir setzen allgemein an:

```
In[•]:= u = {a1, a2, a3} Cos[t];
```

Die Ableitungen sind

```
In[•]:= ut = D[u, t]
Out[•]= {-a1 Sin[t], -a2 Sin[t], -a3 Sin[t]}
```

```
In[•]:= utt = D[u, {t, 2}]
Out[•]= {-a1 Cos[t], -a2 Cos[t], -a3 Cos[t]}
```

Das äußere Produkt ist

```
In[•]:= Cross[ut, utt]
Out[•]= {0, 0, 0}
```

und daher ist natürlich auch der gesuchte Ausdruck gleich null. (Bei einer kreis- oder ellipsenförmigen Bewegung steht der Geschwindigkeitsvektor senkrecht auf den Beschleunigungsvektor!).

Wenn man übrigens einen Term der Form ***c*** f(t) zu u addiert, verschwindet zwar das Kreuzprodukt vielleicht nicht, wohl aber der gesuchte Ausdruck
$\boldsymbol{u}\ .\ [(d\boldsymbol{u}/dt) \times (d^2\boldsymbol{u} / d\,t^2)] = 0$.

```
In[•]:= v = u + {c1, c2, c3} f[t]
Out[•]= {a1 Cos[t] + c1 f[t], a2 Cos[t] + c2 f[t], a3 Cos[t] + c3 f[t]}
```

```
In[•]:= Cross[D[v, t], D[v, {t, 2}]]
Out[•]= {-a3 c2 Cos[t] f′[t] + a2 c3 Cos[t] f′[t] + a3 c2 Sin[t] f″[t] - a2 c3 Sin[t] f″[t],
  a3 c1 Cos[t] f′[t] - a1 c3 Cos[t] f′[t] - a3 c1 Sin[t] f″[t] + a1 c3 Sin[t] f″[t],
  -a2 c1 Cos[t] f′[t] + a1 c2 Cos[t] f′[t] + a2 c1 Sin[t] f″[t] - a1 c2 Sin[t] f″[t]}
```

```
In[•]:= Simplify[v.Cross[D[v, t], D[v, {t, 2}]]]
Out[•]= 0
```

```
In[•]:= ClearAll["Global`*"];
```

7.3

Ein Teilchen bewegt sich entlang der Kurve

$r(t) = \left(2t,\ t^2 + 1,\ t - 1\right).$

Was sind die Komponenten von Geschwindigkeit v(t) und Beschleunigung b(t) in Richtung des Vektors (1, -2, -2)?

Lösungsweg

Wir berechnen die notwendigen Vektoren:

In[•]:=
```
rvec = {2 t, t^2 + 1, t - 1};
vvec = D[rvec, t];
bvec = D[vvec, t];
Print["v = ", vvec];
Print["b = ", bvec];
```

```
v = {2, 2 t, 1}
b = {0, 2, 0}
```

In[•]:=
```
Richtung = {1, -2, -2};
Einheitsvektor = Richtung / Sqrt[Richtung.Richtung]
```

Out[•]= $\left\{\frac{1}{3}, -\frac{2}{3}, -\frac{2}{3}\right\}$

In[•]:=
```
vvec.Einheitsvektor
```

Out[•]= $-\frac{4\,t}{3}$

In[•]:=
```
bvec.Einheitsvektor
```

Out[•]= $-\frac{4}{3}$

In[•]:=
```
ClearAll["Global`*"];
```

7.4

Berechnen Sie in Parameterdarstellung und erläutern Sie die Schnittkurven des Kegels $x^2+y^2=z^2$ mit den Flächen

(a) z=a>0,

(b) z=α y,

(c) x=α,

(d) z+x=α.

Lösungsweg

Eigenlich handelt es sich um einen Doppelkegel mit den Spitzen im Ursprung, den man auch wie folgt parametrisch darstellen kann:

In[•]:=
```
KegelOberfläche = {z Sin[t], z Cos[t], z};
```

In[∘]:=
```
Doppelkegel = ParametricPlot3D[KegelOberfläche, {t, 0, 2 Pi}, {z, -1, 1}]
```

Out[∘]=



(a) z=a > 0

Es ist

$x^2 + y^2 = a^2$

die Kreisgleichung. Eine Parameterdarstellung für solche Kreise ist

r (t) = (a sin (t), a cos (t), a).

Für positive a handelt es sich um die Kreise nur im oberen Teil des Kegels.

In[∘]:=
```
EbeneA = ParametricPlot3D[{x, y, 0.5},
   {x, -1, 1}, {y, -1, 1}, DisplayFunction → Identity];
```

In[•]:=
```
Show[EbeneA, Doppelkegel, DisplayFunction → $DisplayFunction]
```

Out[•]=


(b) z=α y

Das ist eine Ebene durch den Ursprung.

$$x^2 + y^2 = \alpha^2\, y^2 \to x^2 = \left(\alpha^2 - 1\right) y^2 \to x = \pm\sqrt{\alpha^2 - 1}\; y$$

Die Schnittkurven liegen nur für | α|≥1 im Reellen und sind die beiden Geraden:

$$r\,(t) = \left(\pm t\sqrt{\alpha^2 - 1}\ ,\ t,\ \alpha\, t\right) = \left(\pm\sqrt{\alpha^2 - 1}\ ,\ 1,\ \alpha\right) t.$$

In[•]:=
```
EbeneB = ParametricPlot3D[{x, y, 1.5 y},
   {x, -1, 1}, {y, -1, 1}, DisplayFunction → Identity];
```

In[•]:=
```
Show[EbeneB, Doppelkegel, DisplayFunction → $DisplayFunction]
```

Out[•]=



(c) x=α

Das ist eine Ebene parallel zur y-z-Ebene.

$\alpha^2 + y^2 = z^2 \rightarrow z = \pm\sqrt{\alpha^2 + y^2}$ oder $z^2 - y^2 = \alpha^2$

und das ist eine Hyperbel. Eine mögliche Parameterdarstellung ist

$r(t) = (\alpha, \pm\alpha \sinh(t), \pm\alpha \cosh(t))$ mit $t \in \mathbb{R}$.

In[•]:=
```
EbeneC = ParametricPlot3D[{0.5, y, z },
    {y, -1, 1}, {z, -1, 1}, DisplayFunction → Identity];
```

In[•]:=
```
Show[EbeneC, Doppelkegel, DisplayFunction → $DisplayFunction]
```

Out[•]=



(d) z+x=α.

Die Schnittkurve ist eine Parabel,

$x^2 + y^2 = (\alpha - x)^2 \rightarrow y^2 = \alpha^2 - 2\,\alpha\,x,$

und kann in folgende Parameterform gebracht werden:

$$r(t) = \left(x\,,\ \sqrt{\alpha^2 - 2\,\alpha\,x}\,,\ \alpha - x\right) \quad \text{mit} \quad x < \alpha/2.$$

Zur Visualisierung wählen wir α =0.5:

In[•]:=
```
EbeneD = ParametricPlot3D[{x, y, 0.5 - x },
   {x, -1, 1}, {y, -1, 1}, DisplayFunction → Identity];
```

In[•]:= `Show[EbeneD, Doppelkegel, DisplayFunction → $DisplayFunction]`

Out[•]=



In[•]:= `ClearAll["Global`*"];`

7.5

G(x_1, x_2,x_3 ,t) ist differenzierbar in allen vier Variablen; x_1(t), x_2(t) und x_3(t) sind in t differenzierbar. Beweisen Sie die Beziehung (Kontinuitätsgleichung!)

$$\frac{dG}{dt} = \frac{\partial G}{\partial t} + \nabla G . \frac{dr}{dt}$$

Lösungsweg

Wir bilden einfach die totale Ableitung:

In[•]:= `D[G[x1[t], x2[t], x3[t], t], t]`

Out[•]= $G^{(0,0,0,1)}$[x1[t], x2[t], x3[t], t] + x3′[t] $G^{(0,0,1,0)}$[x1[t], x2[t], x3[t], t] + x2′[t] $G^{(0,1,0,0)}$[x1[t], x2[t], x3[t], t] + x1′[t] $G^{(1,0,0,0)}$[x1[t], x2[t], x3[t], t]

Der Nabla Operator ist

In[•]:= `Clear[Nabla]`

In[•]:= `Nabla[xxx_] := {D[xxx, x1], D[xxx, x2], D[xxx, x3]};`

Wir haben hier Nabla neu definiert und nicht das *Mathematica*-Paket (siehe oben) verwendet.

Das ist identisch mit

In[•]:=
```
G^(0,0,0,1)[x1[t], x2[t], x3[t], t] +
  Nabla[G[x1, x2, x3, t]].D[{x1[t], x2[t], x3[t]}, t]
```
Out[•]=
$$G^{(0,0,0,1)}[x1[t], x2[t], x3[t], t] + x3'[t]\, G^{(0,0,1,0)}[x1, x2, x3, t] + x2'[t]\, G^{(0,1,0,0)}[x1, x2, x3, t] + x1'[t]\, G^{(1,0,0,0)}[x1, x2, x3, t]$$

Das stimmt mit der ersten Ableitung (zu Beginn) überein!

In[•]:=
```
ClearAll["Global`*"];
```

7.6

Zeigen Sie zuerst
(a)

$$\kappa = |\dot{r} \times \ddot{r}| / |\dot{r}|^3 = |v \times b| / |v|^3$$

und
(b) berechnen Sie dann κ für die Kurve $\boldsymbol{r}(\varphi) = (3 \cos \varphi, 5 \sin \varphi, 0)$.

Lösungsweg

(a)
Der Tangentenvektor $\boldsymbol{T}$ ist der Einheitsvektor in Richtung des Geschwindigkeit $\boldsymbol{v}/|\boldsymbol{v}|$, die Ableitung von $\boldsymbol{T}$ nach dem Kurvenparameter s ist

$$\kappa \mathbf{H} = d\mathbf{T}/ds$$
$$= (dt/ds)\,(d\boldsymbol{T}/dt)$$
$$= (1/|v|)\,(d\boldsymbol{T}/dt)$$

wobei **H** der Hauptnormalenvektor, ein Einheitsvektor ist. Daraus kann man die Krümmung ablesen.

$$\kappa = 1/|v|\; |d(v/|\boldsymbol{v}|)/dt|$$

Wir berechnen dazu d(v/|v|)/dt:

$$d v_0/dt = d(v/|v|)/dt = (1/|v|)\,(dv/dt) - (v/|v|^3)\, v.b$$
$$= (b - v_0\,(v_0.b))/|v|.$$

Dieser Vektor ($\boldsymbol{b}$ - $\boldsymbol{v_0}$ ($\boldsymbol{v_0}.\boldsymbol{b}$)) zeigt in die Richtung von b, wobei der in Richtung v zeigende Beitrag abgezogen wurde. Sein Absolutbetrag ist also genau der des Kreuzprodukts, wie man folgendermaßen sieht:

$$(b - v_0\,(v_0.b)).(b - v_0\,(v_0.b)) = b.b + (v_0.b)^2 - 2\,(v_0.b)\,(v_0.b) =$$
$$b^2\,(1 - \cos(v_0, b)^2) = |b \times v_0|^2$$

Wir finden daher tatsächlich die Beziehung

$$\kappa = |b \times v_0| / |v|^2 = |b \times v| / |v|^3.$$

Alternativer Lösungsweg:

Da ja $\mathbf{v} = \mathbf{r}' = |\mathbf{v}|\,\mathbf{T} = v\,\mathbf{T}$ gilt, findet man

$$r'' = \frac{d}{dt}(v\,\mathbf{T}) = \frac{dv}{dt}\mathbf{T} + v\,\frac{d\mathbf{T}}{dt}$$

und

$$\frac{d\mathbf{T}}{dt} = \frac{d\mathbf{T}}{ds}\frac{d\mathbf{s}}{dt} = \kappa\, v\, \mathbf{H}$$

Damit gilt

$$r'' = \frac{dv}{dt} T + v^2 \times H$$

und schließlich

$$(r' \times r'') = (vT) \times \left(\frac{dv}{dt} T + v^2 \times H\right) = v^3 \times T \times H = v^3\, \kappa$$

Auflösung nach κ ergibt die gesuchte Beziehung.

(b)

Der Winkel φ ist unser Zeitparameter:

In[•]:=
```
r = {3 Cos[ φ], 5 Sin[ φ], 0};
v = D[r, φ]
vabs = Simplify[Sqrt[v.v]]
```

Out[•]= $\{-3\,\mathrm{Sin}[\varphi],\ 5\,\mathrm{Cos}[\varphi],\ 0\}$

Out[•]= $\sqrt{17 + 8\,\mathrm{Cos}[2\varphi]}$

Der Tangentenvektor ist ein Einheitsvektor:

In[•]:=
```
T = v / vabs
```

Out[•]= $\left\{-\frac{3\,\mathrm{Sin}[\varphi]}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}},\ \frac{5\,\mathrm{Cos}[\varphi]}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}},\ 0\right\}$

In[•]:=
```
kappaH = D[T, φ] / vabs
```

Out[•]= $\left\{\frac{-\frac{3\,\mathrm{Cos}[\varphi]}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}} - \frac{24\,\mathrm{Sin}[\varphi]\,\mathrm{Sin}[2\varphi]}{(17+8\,\mathrm{Cos}[2\varphi])^{3/2}}}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}},\ \frac{-\frac{5\,\mathrm{Sin}[\varphi]}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}} + \frac{40\,\mathrm{Cos}[\varphi]\,\mathrm{Sin}[2\varphi]}{(17+8\,\mathrm{Cos}[2\varphi])^{3/2}}}{\sqrt{17+8\,\mathrm{Cos}[2\varphi]}},\ 0\right\}$

In[•]:=
```
kappa = Sqrt[Simplify[kappaH.kappaH]]
```

Out[•]= $15\sqrt{\frac{1}{(17+8\,\mathrm{Cos}[2\varphi])^3}}$

Damit erhalten wir

$$\mathrm{kappa} = 15 / (17 + 8\,\mathrm{Cos}[2\varphi])^{3/2}$$

als Lösung. Der Krümmungsradius der Ellipsenkurve pendelt also zwischen 3/25 und 9/5.

Mit der unter (a) abgeleiteten Form ergibt sich

In[•]:=
```
Vec = Simplify[Cross[v, D[v, φ]] / vabs^3]
```

Out[•]= $\left\{0,\ 0,\ \frac{15}{(17+8\,\mathrm{Cos}[2\varphi])^{3/2}}\right\}$

also dasselbe kappa wie darüber.

In[∘]:=
```
ClearAll["Global`*"];
```

7.7

(a) Berechnen Sie $\int_B dA\ (x^2+y^2)$; dabei ist B die Oberfläche des Paraboloids z = 2 - (x^2+y^2) über der (x,y)-Ebene.

(b) Berechnen Sie $\int_B dA\ \boldsymbol{n}.\boldsymbol{E}$ für $\boldsymbol{E}$ = (x,y,z); die Fläche B sei die Halbkugel $x^2 + y^2 + z^2 = 1$, z≥ 0 und die Basisfläche $x^2+y^2 \leq 1$, z = 0. Verwenden Sie sowohl kartesische als auch Kugelkoordinaten.

Lösungsweg

(a)

Wir verwenden x, y als Integrationskoordinaten. Auf der Fläche B ist der Ortsvektor

In[∘]:=
```
R = {x, y, 2 - (x^2 + y^2)}
```

Out[∘]= $\{x, y, 2 - x^2 - y^2\}$

Das Flächenelement ist

$dA = | \partial_x \mathbf{R} \times \partial_y \mathbf{R} |\ dx\,dy$

In[∘]:=
```
CrPr = Cross[D[R, x], D[R, y]]
```

Out[∘]= $\{2x, 2y, 1\}$

Damit ist der Betrag

In[∘]:=
```
Sqrt[Sum[CrPr[[i]]^2, {i, 1, 3}]]
```

Out[∘]= $\sqrt{1 + 4x^2 + 4y^2}$

und also

$dA = \sqrt{1 + 4x^2 + 4y^2}\ dx\,dy$.

Die Integration geht über den Bereich $(x^2 + y^2) \leq 2$, da dies die maximalen Werte auf der Integrationsfläche (nämlich bei z=0) sind.

Spätestens hier sehen wir, dass wir mit Zylinderkoordinaten besser daran wären. Wir fangen noch einmal von vorne an und wählen (r,φ,z) als unsere Koordinaten und als Integrationskoordinaten r und φ:

In[∘]:=
```
R = {r Cos[φ], r Sin[φ], 2 - r^2}
```

Out[∘]= $\{r \cos[\varphi], r \sin[\varphi], 2 - r^2\}$

In[∘]:=
```
CrPr = Cross[D[R, r], D[R, φ]]
```

Out[∘]= $\{2 r^2 \cos[\varphi], 2 r^2 \sin[\varphi], r \cos[\varphi]^2 + r \sin[\varphi]^2\}$

In[∘]:=
```
Simplify[Sqrt[Sum[CrPr[[i]]^2, {i, 1, 3}]]]
```
Out[∘]= $\sqrt{r^2 + 4 r^4}$

Es wird also $dA = \sqrt{1 + 4 r^2}\ r\, dr\, d\varphi$ und das Integral läuft über $r^2 \leq 2$:

In[∘]:= 2π `Integrate`$[\sqrt{1+4 r^2}\ r^3, \{r, 0, \sqrt{2}\}]$

Out[∘]= $\frac{149\pi}{30}$

Mit der Hand gerechnet würde man t =
$1 + 4 r^2$ substituieren und dann ein Integral der Form $\int \sqrt{t}\ (t - 1)\, dt$ zu lösen haben.

(b)

Vorbemerkung: Mit etwas Nachdenken kann man das Beispiel ohne Integration lösen. ***E*** ist der Radiusvektor, steht als auf die Kugelfläche immer senkrecht, sein Skalarprodukt mit dem Normalenvektor ***n*** ist daher einfach der Betrag $|r|$, auf der Kugelfläche also 1. Damit bleibt nur das Flächenintegral über die Halbkugelfläche, also 2π. Der Beitrag der Bodenfläche verschwindet, da dort ***n.E***=0.

Da im allgemeinen Fall ***E*** und die Geometrie meist nicht so einfach sind, wollen wir aber dennoch den Standard-Rechengang ausführlich besprechen.

Bei Verwendung kartesischer Koordinaten (x,y als Integrationskoordinaten) ist

$n\, dA = \partial_x \mathbf{R} \times \partial_y \mathbf{R}\ dx\, dy$.

Wir in Beispiel (a) erhalten wir also

In[∘]:= `R = {x, y, Sqrt[1 - (`$x^2 + y^2$`)]}`

Out[∘]= $\{x, y, \sqrt{1 - x^2 - y^2}\}$

In[∘]:=
```
n = Cross[D[R, x], D[R, y]]
```
Out[∘]= $\left\{\frac{x}{\sqrt{1 - x^2 - y^2}}, \frac{y}{\sqrt{1 - x^2 - y^2}}, 1\right\}$

$$n\ dA = \left\{\frac{x}{\sqrt{1 - x^2 - y^2}}, \frac{y}{\sqrt{1 - x^2 - y^2}}, 1\right\} dx\, dy\, .$$

Dieser Vektor zeigt immer in positive z-Richtung (da die z-Komponente konstant den Wert 1 hat).

Der Integrand ist also

In[∘]:=
```
Evec = R;
Integrand = Evec.n
```
Out[∘]= $\frac{x^2}{\sqrt{1 - x^2 - y^2}} + \frac{y^2}{\sqrt{1 - x^2 - y^2}} + \sqrt{1 - x^2 - y^2}$

Das Integral über die Halbkugelfläche ist

$$\int dx\ dy\ \frac{x^2}{\sqrt{1 - x^2 - y^2}} + \frac{y^2}{\sqrt{1 - x^2 - y^2}}$$

$+\sqrt{1-x^2-y^2}$

mit $x^2+y^2 \le 1$

und am besten lösen wir es durch Transformation auf Polarkoordinaten, in welchen es die Form annimmt:

In[•]:=
```
Integrate[2 π / Sqrt[1 - r^2] r, {r, 0, 1}]
```
Out[•]= 2π

Es fehlt noch die Bodenfläche. Dort ist einfach

n dA = {0, 0, 1} dx dy .

(Damit die Integration nach außen zeigt, müssen wir die Richtung umkehren.: n={0,0,-1}.)

Auf der Bodenfläche ist allerdings E.n=0 und so gibt es keinen Beitrag.

Wir rechnen nun alles noch einmal in Kugelkoordinaten. Da auf der Integrationsoberfläche r=1 ist, wählen wir φ und θ als Integrationsvariablen. Es ist

In[•]:=
```
R = { Cos[φ] Sin[θ], Sin[φ] Sin[θ],  Cos[θ] }
```
Out[•]= {Cos[φ] Sin[θ], Sin[θ] Sin[φ], Cos[θ]}

In[•]:=
```
n = Simplify[Cross[D[R, φ], D[R, θ]]]
```
Out[•]= $\{-\mathrm{Cos}[\varphi]\,\mathrm{Sin}[\theta]^2,\ -\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\varphi],\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\}$

$$n\ dA = \{-\mathrm{Cos}[\varphi]\,\mathrm{Sin}[\theta]^2, -\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\varphi], -\frac{1}{2}\,\mathrm{Sin}[2\,\theta]\}\,d\theta\,d\varphi$$

Dieser Vektor zeigt immer in positive z-Richtung.

Der Integrand ist

In[•]:=
```
Evec = R;
Integrand = Simplify[Evec.n]
```
Out[•]= -Sin[θ]

Damit ist die Integration über φ trivial (Faktor 2π) und die über θ ergibt

In[•]:=
```
Integrate[-2 π Sin[θ], {θ, π/2, 0}]
```
Out[•]= 2π

wie bei der Rechnung in kartesischen Koordinaten.

Die Integration über die Bodenfläche wird mit den Variablen r und φ ausgeführt (da ja $\theta=\pi/2$).

In[•]:=
```
R = { r Cos[φ] , r Sin[φ] , 0}
```
Out[•]= {r Cos[φ], r Sin[φ], 0}

In[•]:=
```
n = Simplify[Cross[D[R, r], D[R, φ]]]
```
Out[•]= {0, 0, r}

und wiederum verschwindet

In[•]:=
```
R.n
```
Out[•]= 0

In[•]:=
```
ClearAll["Global`*"];
```

7.8

Bestimmen Sie die geleistete Arbeit, wenn die Kraft

$\boldsymbol{F} = (2x + y,\ x - y^2,\ 0)$

längs der folgenden Wege wirkt:

(a) $x = 0,\ 0 \le y \le 1;\ \ 0 \le x \le 1,\ y = 1$

(b) $x = y,\ 0 \le x \le 1$

(c) $x = 1 - \cos\varphi,\ y = \sin\varphi,$
$0 \le \varphi \le \pi/2$

Lösungsweg

F ist eine konservative Kraft, da

In[•]:=
```
Needs["VectorAnalysis`"]
```

In[•]:=
```
SetCoordinates[Cartesian[x, y, z]]
```
Out[•]= Cartesian[x, y, z]

In[•]:=
```
F = {2 x + y, x - y^2, 0};
Curl[F]
```
Out[•]= {0, 0, 0}

verschwindet. Daher werden alle drei Ergebnisse übereinstimmen. Dennoch wollen wir das explizit nachrechnen.
Es ist jeweils

$$\int (F_x\, dx + F_y\, dy + F_z\, dz)$$

zu berechnen (die Indizes geben Komponenten an). Da die dritte Komponente verschwindet (als auch alle Wege dz=0 haben), brauchen wir nur die beiden ersten Beiträge zu beachten.

(a) 1. Teilstück: dx=0, 2. Teilstück: dy=0.

In[•]:=
```
Inta = Integrate[F[[2]] /. x -> 0, {y, 0, 1}] +
       Integrate[F[[1]] /. y -> 1, {x, 0, 1}]
```
Out[•]= $\frac{5}{3}$

(b) dx=dy

In[•]:=
```
Intb = Integrate[F[[1]] + F[[2]] /. y -> x, {x, 0, 1}]
```
Out[•]= $\frac{5}{3}$

(c) $dx = \sin(\varphi)\, d\varphi$, $dy = \cos(\varphi)\, d\varphi$

In[•]:=
```
Intc = Integrate[Sin[φ] F[[1]] /. {x -> 1 - Cos[φ], y -> Sin[φ]}, {φ, 0, Pi / 2}] +
          Integrate[Cos[φ] F[[2]] /.
                {x -> 1 - Cos[φ], y -> Sin[φ]}, {φ, 0, Pi / 2}]
```

Out[•]= $\frac{2}{3} - \frac{\pi}{4} + \frac{4+\pi}{4}$

Die drei Ergebnisse stimmen also tatsächlich überein.

In[•]:=
```
ClearAll["Global`*"];
```

7.9

Berechnen Sie $\nabla.\nabla\, \phi(x_1, x_2, x_3)$ für
$\phi(x_1, x_2, x_3) = \sin x_1 + x_1^2\, x_2\, x_3$ (vgl. M.7.2).

Lösungsweg

Es ist (siehe M.7.2)

$\nabla\bullet\nabla\ \phi = \triangle\,\phi\ = \phi_{11} + \phi_{22} + \phi_{33} = -\sin x_1\ + 2\ x_2\ x_3$

Umständlicher geht es in Teilschritten:

In[•]:=
```
Needs["VectorAnalysis`"];
```

In[•]:=
```
ϕ = Sin[x1] + x1^2 x2 x3;
G = Grad[ϕ, Cartesian[x1, x2, x3]]
```

Out[•]= $\{2\ x1\ x2\ x3 + \mathrm{Cos}[x1],\ x1^2\ x3,\ x1^2\ x2\}$

In[•]:=
```
Result = Div[G, Cartesian[x1, x2, x3]]
```

Out[•]= $2\ x2\ x3 - \mathrm{Sin}[x1]$

In[•]:=
```
ClearAll["Global`*"];
```

7.10

Man berechne für $|\mathbf{r}|\neq 0$ (vg. M.7.2):
(a) $\Delta(1/|\mathbf{r}|)$
(b) $\Delta \ln|\mathbf{r}|$

Lösungsweg

(a) Es ist

In[•]:=
```
ϕ = 1 / Sqrt[x^2 + y^2 + z^2];
```

und daher

$\triangle\,\phi = \phi_{xx} + \phi_{yy} + \phi_{zz}$

Wir finden die erste Ableitung nach x

In[•]:=
```
D[ϕ, x]
```

Out[•]= $-\frac{x}{\left(x^2+y^2+z^2\right)^{3/2}}$

und die zweite Ableitung nach x

In[•]:=
```
D[ϕ, {x, 2}]
```

Out[•]= $\frac{3\,x^2}{\left(x^2+y^2+z^2\right)^{5/2}}-\frac{1}{\left(x^2+y^2+z^2\right)^{3/2}}$

und analog (einfach durch Permutation) für y und z. Die Summe ist damit

$$\frac{3\ (x^2+y^2+z^2)}{(x^2+y^2+z^2)^{5/2}}-\frac{3}{(x^2+y^2+z^2)^{3/2}}\longrightarrow$$

$$\frac{3}{(x^2+y^2+z^2)^{3/2}}-\frac{3}{(x^2+y^2+z^2)^{3/2}}=0\ \ (\text{für } r\neq 0).$$

(b) Es ist

$$\phi=\mathrm{Log}\left[\sqrt{x^2+y^2+z^2}\right]=\frac{1}{2}\,\mathrm{Log}\left[x^2+y^2+z^2\right];$$

In[•]:=
```
ϕ = 1/2 Log[x^2 + y^2 + z^2];
```

und

$$\triangle\,\phi=\phi_{xx}+\phi_{yy}+\phi_{zz}$$

Wir finden die erste Ableitung nach x

In[•]:=
```
D[ϕ, x]
```

Out[•]= $\frac{x}{x^2+y^2+z^2}$

und die zweite Ableitung nach x

In[•]:=
```
D[ϕ, {x, 2}]
```

Out[•]= $\frac{1}{2}\left(-\frac{4\,x^2}{\left(x^2+y^2+z^2\right)^2}+\frac{2}{x^2+y^2+z^2}\right)$

und analog (einfach durch Permutation) für y und z. Die Summe ist damit

$$\frac{1}{2}\left(-\frac{4\ (x^2+y^2+z^2)}{(x^2+y^2+z^2)^2}+\frac{6}{x^2+y^2+z^2}\right)\longrightarrow$$

$$\frac{1}{2}\left(\frac{2}{x^2+y^2+z^2}\right)=\frac{1}{r^2}.$$

In[•]:=
```
ClearAll["Global`*"];
```

7.12

Berechnen Sie $\nabla(\nabla.\boldsymbol{u})$ für den Einheitsvektor $\boldsymbol{u}=\boldsymbol{r}/|\boldsymbol{r}|$.

Lösungsweg

Die Rechnung ergibt

```
In[•]:= Needs["VectorAnalysis`"];
```

```
In[•]:= r = {x, y, z};
SetCoordinates[Cartesian[x, y, z]];
Simplify[Grad[Div[r / Sqrt[r.r]]]]
```

Out[•]= $\left\{-\frac{2x}{(x^2+y^2+z^2)^{3/2}}, -\frac{2y}{(x^2+y^2+z^2)^{3/2}}, -\frac{2z}{(x^2+y^2+z^2)^{3/2}}\right\}$

also - $\boldsymbol{r}/(|\boldsymbol{r}|)^3$.

Wie sieht das in sphärischen Polarkoordinaten aus?

```
In[•]:= r = {R, 0, 0};
SetCoordinates[Spherical[R, θ, φ]];
Simplify[Grad[Div[r / Sqrt[r.r]]]]
```

Out[•]= $\left\{-\frac{2\sqrt{R^2}}{R^3}, 0, 0\right\}$

also ebenfalls - $\boldsymbol{r}/(|\boldsymbol{r}|)^3$.

```
In[•]:= ClearAll["Global`*"];
```

7.14

Ein Vektorfeld habe die Form

$a(x, y, z) = (x + \alpha y, y + \beta x, z)$.

Bestimmen Sie α und β so, dass $\textbf{rot}\ \boldsymbol{a} = 0$ gilt, und berechnen Sie das entsprechende Potenzialfeld.

Lösungsweg

Wir setzen an

```
In[•]:= Needs["VectorAnalysis`"];
```

```
In[•]:= a = {x + alpha y, y + beta x, z};
SetCoordinates[Cartesian[x, y, z]];
Curl[a]
```

Out[•]= {0, 0, -alpha + beta}

Für $\alpha = \beta$ handelt es sich daher um eine konservative Kraft.
Es ist a=-grad Φ, daher

$\partial_x \Phi = -x - \text{alpha}\, y$

$\partial_y \Phi = -y - \text{alpha}\, x$

$\partial_z \Phi = -z$

$\partial_x \Phi = -x - \text{alpha}\, y \rightarrow$

$\Phi = -x^2/2 - x\, y\, \text{alpha} + c(y, z)$

$\partial_y \Phi = -y - \text{alpha}\, x = -x\, \text{alpha} + \partial_y c(y, z) \to$

$\partial_y c(y, z) = -y \to c(y, z) = -y^2/2 + c(z)$

$\partial_z \Phi = -z = \partial_z c(z) \to c(z) = -z^2/2 + c$

und daraus schließlich

$\Phi(x.y.z) = -(x^2 + y^2 + z^2)/2 - \text{alpha}\, x\, y + c.$

```
In[•]:= ClearAll["Global`*"];
```

7.16

Zeigen Sie, dass die Kraft

$\boldsymbol{F}(x, y, z) = (y^3,\ 3\, x\, y^2 + 2\, y \cos z,\ -y^2 \sin z)$

eine konservative Kraft ist, und bestimmen Sie das entsprechende Potenzial.

Lösungsweg

Die Kraft ist

```
In[•]:= F = {y^3, 3 x y^2 + 2 y Cos[z], -y^2 Sin[z]}
Out[•]= {y^3, 3 x y^2 + 2 y Cos[z], -y^2 Sin[z]}
```

```
In[•]:= Needs["VectorAnalysis`"];
SetCoordinates[Cartesian[x, y, z]];
Curl[F]
Out[•]= {0, 0, 0}
```

Der Rotor verschwindet und daher ist die Kraft konservativ und es gilt
F=-grad Φ.

$\partial_x \Phi = -y^3 \to \Phi = -x\, y^3 + a(y, z)$

$\partial_y \Phi = -3\, x\, y^2 - 2\, y \cos z$

$= -3\, x\, y^2 + \partial_y a(y, z)$

$\to \partial_y a(y, z) = -2\, y \cos z$

$\to a(y, z) = -y^2 \cos z + b(z)$

$\partial_z \Phi = y^2 \sin z = \partial_z a(y, z)$

$= y^2 \sin z + \partial_z b(z)$

$\to \partial_z b(z) = 0 \to b(z) = b$

$\to \Phi(x, y, z) = -x\, y^3 - y^2 \cos z + b$

...und damit die gesuchte Lösung.

```
In[•]:= ClearAll["Global`*"];
```

7.18

Die Gravitationskraft hat die Form

$$\boldsymbol{F}(r) = -\frac{G\, M\, m\, \mathbf{r}}{(|\mathbf{r}|)^3}.$$

Zeigen Sie, dass es sich hierbei um eine konservative Kraft handelt.

Lösungsweg

Die Kraft ist

In[•]:= F = -G M m {x, y, z} / Sqrt[$x^2 + y^2 + z^2$]^3

Out[•]= $\left\{-\frac{G\,m\,M\,x}{(x^2+y^2+z^2)^{3/2}}, -\frac{G\,m\,M\,y}{(x^2+y^2+z^2)^{3/2}}, -\frac{G\,m\,M\,z}{(x^2+y^2+z^2)^{3/2}}\right\}$

Der Rotor kann auch durch die Determinante folgender Matrix bestimmt werden:

$$\begin{pmatrix} e_x & e_{yx} & e_z \\ \partial_x & \partial_y & \partial_z \\ \frac{x}{(x^2+y^2+z^2)^{3/2}} & \frac{y}{(x^2+y^2+z^2)^{3/2}} & \frac{z}{(x^2+y^2+z^2)^{3/2}} \end{pmatrix}$$

Da wir schon öfter den Rotor von **r**/r^3berechnet haben

In[•]:=
```
Needs["VectorAnalysis`"];
Curl[F, Cartesian[x, y, z]]
```

Out[•]= {0, 0, 0}

Der Rotor verschwindet und daher ist die Kraft konservativ und es gilt
F=-grad Φ.

In[•]:=
```
ClearAll["Global`*"];
```

7.20

Bilden Sie Rotor und Divergenz des Vektorfeldes

$F\,(x_1, x_2, x_3) = e^{x_1+x_2+x_3}\,(x_2\,x_3\,(1+x_1),\; x_1\,x_3\,(1+x_2),\; x_1\,x_2\,(1+x_3))$

Lösungsweg

Wir verwendet (der einfacheren Notation zuliebe) x, y, und z:

In[•]:= F = e^{x+y+z} {y z (1 + x), x z (1 + y), x y (1 + z)};

Die Divergenz ist

$\partial_x F_x + \partial_y F_y + \partial_z F_z$

und daher

In[•]:= D1 = ∂_x F[[1]]

Out[•]= $e^{x+y+z}\,y\,z + e^{x+y+z}\,(1+x)\,y\,z$

In[•]:= D2 = ∂_y F[[2]]

Out[•]= $e^{x+y+z}\,x\,z + e^{x+y+z}\,x\,(1+y)\,z$

In[•]:= D3 = ∂_z F[[3]]

Out[•]= $e^{x+y+z}\,x\,y + e^{x+y+z}\,x\,y\,(1+z)$

In[•]:= `Simplify[D1 + D2 + D3]`

Out[•]= $e^{x+y+z}\,(2\,x\,z+2\,y\,z+x\,y\,(2+3\,z))$

In[•]:= `DivergenzF = Apart[%]`

Out[•]= $e^{x+y+z}\,(2\,x\,y+2\,x\,z+2\,y\,z+3\,x\,y\,z)$

Der Rotor ist das Kreuzprodukt von ∇ mit F. In Komponentes ist dies

In[•]:= `RotorF1 = ∂y F[[3]] - ∂z F[[2]]`

Out[•]= $-e^{x+y+z}\,x\,(1+y)-e^{x+y+z}\,x\,(1+y)\,z+e^{x+y+z}\,x\,(1+z)+e^{x+y+z}\,x\,y\,(1+z)$

In[•]:= `RotorF2 = -(∂x F[[3]] - ∂z F[[1]])`

Out[•]= $e^{x+y+z}\,(1+x)\,y+e^{x+y+z}\,(1+x)\,y\,z-e^{x+y+z}\,y\,(1+z)-e^{x+y+z}\,x\,y\,(1+z)$

In[•]:= `RotorF3 = ∂x F[[2]] - ∂y F[[1]]`

Out[•]= $-e^{x+y+z}\,(1+x)\,z-e^{x+y+z}\,(1+x)\,y\,z+e^{x+y+z}\,(1+y)\,z+e^{x+y+z}\,x\,(1+y)\,z$

MATHEMATICA erkennt noch nicht, dass diese Ausdrücke jeweils null sind, wir müssen dazu die Funktion "Simplify" nutzen:

In[•]:= `Simplify[{RotorF1, RotorF2, RotorF3}]`

Out[•]= `{0, 0, 0}`

Alternativ können wir auch einfach die vorgefertigten Funktionen verwenden:

In[•]:= `Needs["VectorAnalysis`"];`

In[•]:= `Curl[F, Cartesian[x, y, z]]`

Out[•]= $\{-e^{x+y+z}\,x\,(1+y)-e^{x+y+z}\,x\,(1+y)\,z+e^{x+y+z}\,x\,(1+z)+e^{x+y+z}\,x\,y\,(1+z),$
$e^{x+y+z}\,(1+x)\,y+e^{x+y+z}\,(1+x)\,y\,z-e^{x+y+z}\,y\,(1+z)-e^{x+y+z}\,x\,y\,(1+z),$
$-e^{x+y+z}\,(1+x)\,z-e^{x+y+z}\,(1+x)\,y\,z+e^{x+y+z}\,(1+y)\,z+e^{x+y+z}\,x\,(1+y)\,z\}$

In[•]:= `Simplify[%]`

Out[•]= `{0, 0, 0}`

In[•]:= `Div[F, Cartesian[x, y, z]]`

Out[•]= $e^{x+y+z}\,x\,y+e^{x+y+z}\,x\,z+e^{x+y+z}\,y\,z+e^{x+y+z}\,(1+x)\,y\,z+e^{x+y+z}\,x\,(1+y)\,z+e^{x+y+z}\,x\,y\,(1+z)$

In[•]:= `Apart[%]`

Out[•]= $e^{x+y+z}\,(2\,x\,y+2\,x\,z+2\,y\,z+3\,x\,y\,z)$

In[•]:= `ClearAll["Global`*"];`

7.21

Sei f(r) eine differenzierbare Funktion (r=|***r***|). Berechnen Sie **∇×** (***r*** f(r)).

Lösungsweg

Der wesentliche konzeptuelle Punkt ist hier: Es handelt sich um eine Zentralkraft! Daher wird der Rotor sicher verschwinden. Wir können das durch explizite Berechnung überprüfen. Dazu schreiben wir den Betrag des Ortsvektors als Funktion der Ortskoordinaten an und differenzieren dann. In Mathematica folgen wir der oben (vor der Beispielliste) angegebenen Vorgangsweise.

```
In[]:= Needs["VectorAnalysis`"];
In[]:= SetCoordinates[Cartesian[x1, x2, x3]];
In[]:= r = Sqrt[x1^2 + x2^2 + x3^2]
Out[]= √(x1² + x2² + x3²)
In[]:= Curl[{x1, x2, x3} fun[r]]
Out[]= {0, 0, 0}
```

Bei der Rechnung per Hand (bzw. Kopf) muss man beachten, dass auch die Funktion (hier fun[r]) nach den Koordinaten abgeleitet werden muss!

```
In[]:= ClearAll["Global`*"];
```

7.22

Man zeige, dass

$2\,x\,y^3\,dx + 3\,x^2\,y^2\,dy - (z\cos z + \sin z)\;dz$

das exakte Differenzial eines Potenzials Φ(x,y,z) ist. Wie lautet Φ?

Lösungsweg

Ein exaktes Differenzial hat die Form grad Φ .d ***r***, daher lesen wir die Komponenten ab:

```
In[]:= GradPhi = {2 x y^3, 3 x^2 y^2, -z Cos[z] - Sin [z]}
Out[]= {2 x y^3, 3 x^2 y^2, -z Cos[z] - Sin[z]}
```

Falls das der Gradient eines Potentials ist (also eine konservative Kraft), dann muss dessen Rotor verschwinden:

```
In[]:= Needs["VectorAnalysis`"];
In[]:= Curl [GradPhi, Cartesian[x, y, z]]
Out[]= {0, 0, 0}
```

Es handelt sich also in der Tat um ein Potential. Wir bestimmen Φ aus der Integration der drei Gleichungen

$\partial_x \Phi = 2\,x\,y^3$

$\partial_y \Phi = 3\,x^2\,y^2$

$\partial_z \Phi = -z\,\mathrm{Cos}[z] - \mathrm{Sin}\,[z]$

$\partial_x \Phi = 2\,x\,y^3 \rightarrow\ \mathrm{Phi} = x^2\,y^3 + a\,(y,\,z)$

$\partial_y \Phi = 3\,x^2\,y^2$

$\qquad = 3\,x^2\,y^2 + \partial_y a\,(y,\,z)$

$\rightarrow\ \partial_y a\,(y,\,z) = 0$

$\rightarrow a\,(y,\,z) = a\,(z)$

$\partial_z \Phi = -z\,\mathrm{Cos}[z] - \mathrm{Sin}\,[z]$

$= \partial_z a(z)$

$\to a(z) = -z\,\mathrm{Sin}[z] + a$

und daher

$\Phi(x, y, z) = x^2 y^3 - z\,\mathrm{Sin}[z] + a$

In[∘]:= `ClearAll["Global`*"];`

7.23

Berechnen Sie für die Kraft **F** das Wegintegral

$g(x_0, y_0, a) = \int d\mathbf{r}\,.\,\mathbf{F}$

entlang des geschlossenen Weges in der x-y Ebene

$(x_0, y_0) \to (x_0 + a, y_0) \to (x_0 + a, y_0 + a) \to (x_0, y_0 + a) \to (x_0, y_0)$.

Vergleichen Sie dann

$\lim_{a\to 0} g(x_0, y_0, a) / a^2$

mit rot **F**; nutzen Sie dazu die Taylorreihe für die Komponenten von F. Was fällt Ihnen auf?

Lösungsweg

Da wie den Limes kleiner a betrachten, verwenden wir die Taylorentwicklung für die Komponenten von F:

$F_j(x_0 + s, y_0 + t, 0) =$
$F_j(x_0, y_0, 0) + \partial_x F_j(x_0, y_0, 0)\, s + \partial_y F_j(x_0, y_0, 0)\, t + O(s)\, O(t)$

Die vier Integrale entlang der Kanten des Quadrats (gegen den Uhrzeiger) ergeben

$\int_{x_0}^{x_0+a} ds\, F_x(x_0 + s, y_0, 0) =$
$\int_{x_0}^{x_0+a} ds\, (F_x(x_0, y_0, 0) + \partial_x F_x(x_0, y_0, 0)\, s) =$
$F_x(x_0, y_0, 0)\, a + \partial_x F_x(x_0, y_0, 0)\, a^2/2 + O(a^3)$

$\int_{y_0}^{y_0+a} dt\, F_y(x_0 + a, y_0 + t, 0) =$
$\int_{y_0}^{y_0+a} dt\, (F_y(x_0, y_0, 0) + \partial_x F_y(x_0, y_0, 0)\, a$
$+ \partial_y F_y(x_0, y_0, 0)\, t) =$
$F_y(x_0, y_0, 0)\, a + \partial_x F_y(x_0, y_0, 0)\, a^2 + \partial_y F_y(x_0, y_0, 0)\, a^2/2 + O(a^3)$

$\int_{x_0+a}^{x_0} ds\, F_x(x_0 + s, y_0 + a, 0) =$
$\int_{x_0+a}^{x_0} ds\, (F_x(x_0, y_0, 0) + \partial_x F_x(x_0, y_0, 0)\, s + \partial_y F_x(x_0, y_0, 0)\, a) =$
$-F_x(x_0, y_0, 0)\, a - \partial_x F_x(x_0, y_0, 0)\, a^2/2 - \partial_y F_x(x_0, y_0, 0)\, a^2 + O(a^3)$

$\int_{y_0}^{y_0+a} dt\, F_y(x_0, y_0 + t, 0) =$
$\int_{y_0+a}^{y_0} dt\, (F_y(x_0, y_0, 0) + \partial_y F_y(x_0, y_0, 0)\, t) =$

$-F_y(x_0, y_0, 0)\, a - \partial_y F_y(x_0, y_0, 0)\, a^2/2 + O(a^3)$

(Die Indizes von F bezeichnen hier Komponenten, nicht Ableitungen.)

Die Summe ergibt

$g(x_0, y_0, a) =$
$(\partial_x F_y(x_0, y_0, 0) - \partial_y F_x(x_0, y_0 0))\, a^2 + O(a^3)$

Daher ergibt sich der Grenzwert

$\lim_{a\to 0} g(x_0, y_0, a)/a^2 = \partial_x F_y(x_0, y_0, 0) - \partial_y F_x(x_0, y_0, 0)$

Das ist genau die z-Komponente des Rotors (rot $F)_z$.

Man kann den Rotor daher als Beitrag einer Integration um den Rand einer Fläche pro Flächeneinhiet ansehen. Diese Beobachtung wird in Kapitel 9 durch den Satz von Stokes bestätigt.

In[∘]:=
```
ClearAll["Global`*"];
```

7.A.1

Information zu Koordinatensystemen

In Mathematica kann das Koordinatensystem festgelegt werden, damit man dann den Gradienten (Grad), die Divergenz (Div), den Rotor (Curl) und den Laplace-Operator (Laplacian) berechnen kann.

In[∘]:=
```
Needs["VectorAnalysis`"];
```

Man kann dann zum Beispiel folgende Befehle ausführen, um den Rotor eines Vektors zu berechnen:

In[∘]:=
```
SetCoordinates[Cartesian[x1, x2, x3]];
```

In[∘]:=
```
Curl[{x1 x2 , x2 x3 , x1 x3 }]
```

Out[∘]=
```
{-x2, -x3, -x1}
```

Da dabei der Rechengang natürlicherweise nicht angegeben wird, werden in unseren Lösungsbeschreibungen Rechenschritte und Beispiele, die einfach nur die Berechnung dieser Differenzialoperationen sind, nicht angeben.

In[∘]:=
```
ClearAll["Global`*"];
```